Wolfram Mathematica

Informatika 1, 13. előadás

Csikja Rudolf notebook-jai alapján.

Függvények

Jelölések

In[1]:=
```
Clear[f]
f@5
5 // f
f[5]
f@{1, 5, 3}
{1, 5, 3} // f
f[{1, 5, 3}]
```

Out[2]= `f[5]`

Out[3]= `f[5]`

Out[4]= `f[5]`

Out[5]= `f[{1, 5, 3}]`

Out[6]= `f[{1, 5, 3}]`

Out[7]= `f[{1, 5, 3}]`

Függvények definiálása

Pontonkénti definíció

In[8]:=
```
Clear[f]
f[0] := 1
f[1] := 3
```

In[11]:=
```
? f
```

```
Global`f
```

```
f[0] := 1

f[1] := 3
```

In[12]:=
```
f[]
f[12]
f[0]
f[Range[3]]
```
Out[12]= f[]

Out[13]= f[12]

Out[14]= 1

Out[15]= f[{1, 2, 3}]

Általános definíció

In[16]:=
```
Clear[f]
f[x_] := x + 1
```

In[18]:=
```
? f
```

Global`f

f[x_] := x + 1

In[19]:=
```
f[0]
f[1.5]
f[x, y]
```
Out[19]= 1

Out[20]= 2.5

Out[21]= f[x, y]

Vegyes definíció

In[22]:=
```
Clear[f]
f[x_] := x + 1
f[0] := 100
```

In[25]:=
```
f[0]
f[23]
```
Out[25]= 100

Out[26]= 24

Többváltozós függvények

In[27]:=
```
ClearAll[f, g]
f[x_, y_] := x + y
g[a_, b_, c_] := a + b^c
```

In[30]:=
```
f[a^x, b]
g[b, a, x]
```
Out[30]= $a^x + b$

Out[31]= $a^x + b$

Az argumentum típusa

In[32]:=
```
Head[1]
Head[1/2]
Head[1.2]
Head[{1, 2, 3}]
```
Out[32]= Integer

Out[33]= Rational

Out[34]= Real

Out[35]= List

In[36]:=
```
Clear[f]
f[x_Integer] := x + 1
f[x_Real] := x + 2
f[x_Rational] := x + 3
f[x_List] := x + 4
```

In[41]:=
```
f[1]
f[1.2]
f[1/2]
f[{1, 2, 3}]
```
Out[41]= 2

Out[42]= 3.2

Out[43]= $\frac{7}{2}$

Out[44]= {5, 6, 7}

Anonímusz függvények

Function

Elnevezhetjük a függényt, a két függvény definíció ekvivalens.

In[45]:=
```
Clear[f]
f[x_] := x^2
f = Function[{x}, x^2]
```
Out[47]= Function[{x}, x^2]

In[48]:=
```
f[3]
f[a + b]
```
Out[48]= 9

Out[49]= $(a + b)^2$

Vagy elnevezés nélkül is használhatjuk. (A másik megadással ez nem működne.)

In[50]:=
```
Function[{x}, 3 x^2 + 5][12 + a]
```
Out[50]= $5 + 3 (12 + a)^2$

Tiszta (pure) függvények

Itt a változónak nem adunk nevet.

In[51]:=
```
Clear[f]
f = (#^2 + 1) &
```
Out[52]= $\#1^2 + 1$ &

Ha akarjuk még a függvénynek sem kell nevet adni.

In[53]:=
```
Clear[f]
(#^2 + 1) &[10]
```
Out[54]= 101

Több változó esetében.

In[55]:=
```
Clear[f]
f = (5 #1 + #2^2)^#3 &;
```
In[57]:=
```
f[a, b, c]
```
Out[57]= $(5 a + b^2)^c$

Magasabb szintű listafüggvények

Map

In[58]:=
```
Sin[{a, b, c, d}]
Power[{a, b, c, d}, 2]
```
Out[58]= {Sin[a], Sin[b], Sin[c], Sin[d]}

Out[59]= $\{a^2, b^2, c^2, d^2\}$

```
In[60]:= Map[Sin, {a, b, c, d}]
Out[60]= {Sin[a], Sin[b], Sin[c], Sin[d]}

In[61]:= Clear[f]
         f[x_] := x^2;
         Map[f, {a, b, c}]
Out[63]= {a^2, b^2, c^2}

In[64]:= Map[#^2 &, {a, b, c}]
Out[64]= {a^2, b^2, c^2}

In[65]:= Select[RandomInteger[100, 100], 10 < # < 70 &]
Out[65]= {22, 60, 59, 51, 17, 54, 57, 21, 55, 68, 33, 14, 57, 23, 33, 39, 30, 27,
          37, 13, 53, 27, 62, 36, 60, 63, 43, 57, 67, 52, 49, 29, 32, 66, 45, 11, 56,
          37, 38, 49, 27, 25, 11, 47, 19, 45, 13, 15, 34, 19, 46, 21, 11, 38, 68, 53}
```

Más írásmóddal:

```
In[66]:= #^2 & /@ {a, b, c}
Out[66]= {a^2, b^2, c^2}
```

Apply

```
In[67]:= Apply[f, {a, b, c, d}]
Out[67]= f[a, b, c, d]

In[68]:= Apply[Plus, {a, b, c, d}]
Out[68]= a + b + c + d
```

Más írásmóddal:

```
In[69]:= Times @@ {a, b, c}
Out[69]= a b c
```

Rekurzió

Kezdeti tagokkal megadva

Egy elemre visszautalva:

$a_0 = 2$

$a_n = 2\,a_{n-1} - 1$

```
In[70]:= a[0] := 2
         a[n_] := 2 a[n - 1] - 1
```

In[72]:=
```
a[2]
Table[a[i], {i, 1, 10}]
```

Out[72]= 5

Out[73]= {3, 5, 9, 17, 33, 65, 129, 257, 513, 1025}

Több elemre visszautalva:

$b_0 = 1$

$b_1 = 3$

$b_2 = -2$

$b_n = b_{n-1} b_{n-2} + b_{n-1} b_{n-3}$

In[74]:=
```
b[0] := 1
b[1] := 3
b[2] := -2
b[n_] := b[n - 1] b[n - 2] + b[n - 1] b[n - 3]
```

In[78]:=
```
b[4]
Table[b[i], {i, 1, 10}]
```

Out[78]= -8

Out[79]= {3, -2, -8, -8, 80, -1280, -92 160, 110 592 000, -10 333 716 480 000, -1 141 874 017 645 363 200 000}

Feltétellel megadva

Az első öttel osztható számig adjuk össze a számokat

$$c_n := \begin{cases} n + c_{n-1}, & \text{ha } n \text{ nem osztható öttel} \\ 0, & \text{ha } n \text{ osztható öttel} \end{cases}$$

In[80]:=
```
c[n_] := If[Mod[n, 5] == 0, 0, n + c[n - 1]]
```

In[81]:=
```
c[8]
Table[c[i], {i, 1, 20}]
```

Out[81]= 21

Out[82]= {1, 3, 6, 10, 0, 6, 13, 21, 30, 0, 11, 23, 36, 50, 0, 16, 33, 51, 70, 0}

Egyenletmegoldás, szabályok

In[83]:=
```
Solve[x + 2 == 5]
```

Out[83]= {{x → 3}}

In[84]:=
```
a → b
```

Out[84]= a → b

Szabály alkalmazása

In[85]:=
```
Clear[f]
ReplaceAll[x + 2 y + z, x → y + a]
```

Out[86]= $a + 3 y + z$

Rövidebben:

In[87]:= x^2 /. x → (a + b)

Out[87]= $(a + b)^2$

In[88]:=
```
Solve[x^2 + 2 x - 11 == 0, x]
Solve[x^2 + 2 x - 11 == 0, x] /. x → 5
x /. Solve[x^2 + 2 x - 11 == 0, x]
```

Out[88]= $\{\{x \to -1 - 2\sqrt{3}\}, \{x \to -1 + 2\sqrt{3}\}\}$

Out[89]= $\{\{5 \to -1 - 2\sqrt{3}\}, \{5 \to -1 + 2\sqrt{3}\}\}$

Out[90]= $\{-1 - 2\sqrt{3}, -1 + 2\sqrt{3}\}$

In[91]:=
```
x^2 + 2 x - 11 /. Solve[x^2 + 2 x - 11 == 0, x]
x^2 + 2 x - 11 /. Solve[x^2 + 2 x - 11 == 0, x] // Simplify
```

Out[91]= $\{-11 + 2\left(-1 - 2\sqrt{3}\right) + \left(-1 - 2\sqrt{3}\right)^2, -11 + 2\left(-1 + 2\sqrt{3}\right) + \left(-1 + 2\sqrt{3}\right)^2\}$

Out[92]= $\{0, 0\}$

Integrálás, deriválás

In[93]:=
D[x^2, x]
Integrate[x^2, x]

Out[93]= $2 x$

Out[94]= $\frac{x^3}{3}$

Grafika

Grafikus primitívek

Ez egy üres grafika.

In[95]:=
```
Framed[Graphics[]]
```

Out[95]=



A grafikán belül egy listában adhatjuk meg grafikus primitíveket.

Pont

Egy 2 dimenziós pont reprezentációja (Descarte koordinátákban).

In[96]:=
```
Point[{1, 2}]
```

Out[96]=
```
Point[{1, 2}]
```

Pontok halmaza.

In[97]:=
```
Point[{{0, 0}, {1, 2}, {-3, 5}}]
```

Out[97]=
```
Point[{{0, 0}, {1, 2}, {-3, 5}}]
```

In[98]:=
```
Graphics@Point[{0, 0}] // Framed
```

Out[98]=



In[99]:=
```
Graphics[{
   Red, PointSize[0.1],
   Point[{0, 0}],
   Blue, PointSize[0.05],
   Point[{{1, 2}, {-3, 5}, {-1, -1}, {-2, 0}}]
  }, PlotRange → {{-5, 3}, {-2, 10}}] // Framed
```

Out[99]=



In[100]:=
```
Graphics@{
  PointSize[Large], Red,
  Point@Table[{Sin[t], Cos[t]}, {t, 0, 2 π, 2 π/12}]
 }
```

Out[100]=



Szakasz, törött vonal

Két pontot összekötő egyenes szakasz.

In[101]:=
```
Line[{{0, 0}, {1, 1}}]
Graphics@{
  Thick,
  Line[{{0, 0}, {1, 1}}],
  Red, PointSize[0.05],
  Point[{{0, 0}, {1, 1}}]
 }
```

Out[101]= `Line[{{0, 0}, {1, 1}}]`

Out[102]=



Több pontot összekötő törött vonal.

In[103]:=
```
points = {{0, 0}, {1, 1}, {1, 2}, {3, 1}, {2, 0}};
Graphics@{
  Thick,
  Line[points],
  Red, PointSize[0.05],
  Point[points]
 }
Clear[points]
```

Out[104]=



In[106]:=
```
Framed@Graphics[
  {Red, Line[{{0, -1}, {-2, 1}, {2, -1}}],
   Blue, Thick, Line[{{-3, -2}, {2, 1}}]
  }
 ]
```

Out[106]=



In[107]:=
```
line1 = RandomReal[{-1, 1}, {10, 2}];
line2 = RandomReal[{-1, 1}, {50, 2, 2}];
Framed@Graphics[{
   Thick, Red, Line@line1,
   Thickness[Medium], Blue, Line@line2}]
ClearAll[line1, line2]
```

Out[109]=



Kör, elipszis

In[111]:=

```
Graphics@Circle[]
```

Out[111]=



In[112]:=
```
Graphics[{
  Circle[{-1, -1}, {2, 1}],
  Red,
  Circle[{0, 0}, 1, {0, 5 π / 4}],
  Blue, Thick,
  Circle[{1, 2}, 0.6]
 }]
```

Out[112]=



Téglalap

In[113]:=
```
Options[Rectangle]
```

Out[113]=
```
{RoundingRadius → 0.}
```

In[114]:=
```
Graphics@Rectangle[{0, 0}, {2, 1}]
Graphics@Rectangle[{0, 0}, {2, 1}, RoundingRadius → 0.1]
```

Out[114]=

Out[115]=

In[116]:=
```
Graphics[{
  Red, Rectangle[{0, 0}, {1, 2}],
  Opacity[0.5], Blue, Rectangle[{-2, -3}, {2, 1}, RoundingRadius → 1],
  Opacity[1], FaceForm[Green],
  EdgeForm[{Dashed, Black, Thick}], Rectangle[{-3, 3}, {-1, -2}]
 }]
```

Out[116]=



Körlemez

In[117]:=
```
Graphics@Disk[]
```

Out[117]=


In[118]:=
```
Graphics@{
  Orange,
  Disk[{3, 0}, {1, 2}],
  Opacity[0.5],
  EdgeForm[{Dashed, Thick, Red}], FaceForm[Blue],
  Disk[{0, 0}, 1, {0, 5 π / 4}]
 }
```

Out[118]=



Sokszög

In[119]:=
```
Graphics@Polygon[{{0, 0}, {1, 0}, {2, 1}, {1, 3}}]
```

Out[119]=



In[120]:=
```
Clear[ngon]
ngon[n_Integer, f_: Polygon] := Graphics@{
   Brown,
   f@Table[{Sin[t], Cos[t]}, {t, 0, 2 π, 2 π / n}]
  }
```

In[122]:= **ngon[5]**

Out[122]=



In[123]:= **ngon[70, Line]**

Out[123]=



```
In[124]:= ngon /@ Range[3, 8]
```

Out[124]=



```
In[125]:= GraphicsRow[ngon /@ Range[3, 8], ImageSize → {Automatic, 100}]
```

Out[125]=



Szöveg

In[126]:=
```
Graphics[{Text["Hello"], Text["World!", {1, 1}], Text[
    Style["Alma", FontColor → Red, FontSize → 22], {-1, -1}]}, PlotRange → 2] // Framed
```

Out[126]=



Színek

Beépített színek nevek szerint

In[127]:=
```
? Red
Green
Blue
Cyan
Magenta
Yellow
Black
White
Orange
Purple
Brown
Pink
```

Red represents the color red in graphics or style specifications. >>

Out[128]=

Out[129]=

Out[130]=

Out[131]=

Out[132]=

Out[133]=

Out[134]=

Out[135]=

Out[136]=

Out[137]=

Out[138]=

In[139]:=
```
FullForm[Red]
```

Out[139]//FullForm=
```
RGBColor[1, 0, 0]
```

Beépített színskálák

In[140]:=
```
ColorData["TemperatureMap"][0.5]
```

Out[140]=

Színmodellek

In[141]:=
```
GrayLevel[level, opacity]
RGBColor[red, green, blue, opacity]
Hue[hue, saturation, brigthness, opacity]
CMYKColor[cian, magenta, yellow, black]
```

Out[141]= `GrayLevel[level, opacity]`

Out[142]= `RGBColor[red, green, blue, opacity]`

Out[143]= `Hue[hue, saturation, brigthness, opacity]`

Out[144]= `CMYKColor[cian, magenta, yellow, black]`

In[145]:=
```
Graphics[{GrayLevel[0.8], Disk[]}]
```

Out[145]=

In[146]:= `Graphics[{RGBColor[1, 0.3, 0], Disk[]}]`

Out[146]=



In[147]:= `Graphics[{Hue[0.1, 0.8, 1], Disk[]}]`

Out[147]=



In[148]:= `IdentityMatrix[4]`

Out[148]= `{{1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 1}}`

In[149]:=
```
Graphics[
 MapThread[{CMYKColor[#1], Disk[{#2, 0}, 0.4]} &, {IdentityMatrix[4], Range[4]}],
 ImageSize → 400]
```

Out[149]=



Magasabb szintű grafikus függvények

In[150]:=
```
Graphics@Line@Table[{x, Sin[x]}, {x, -2 π, 2 π, 0.1}]
```

Out[150]=



Valós egyváltozós függvény ábrázolása

In[151]:=
```
Plot[Sin[x], {x, -2 π, 2 π}]
```

Out[151]=



In[152]:= `Plot[Sin[x], {x, -2 π, 2 π}, PlotStyle → {Thick, Red}, GridLines → Automatic, GridLinesStyle → Dotted]`

Out[152]=


Valós kétváltozós függvények ábrázolása

In[153]:= `Plot3D[(x + y) Sin[x - y], {x, -5, 5}, {y, -5, 5}]`

Out[153]=


Lista elemeinek ábrázolása

In[154]:=
```
Sin[Range[0, 2 π, 0.1]]
```

Out[154]=
```
{0., 0.0998334, 0.198669, 0.29552, 0.389418, 0.479426, 0.564642, 0.644218,
 0.717356, 0.783327, 0.841471, 0.891207, 0.932039, 0.963558, 0.98545,
 0.997495, 0.999574, 0.991665, 0.973848, 0.9463, 0.909297, 0.863209,
 0.808496, 0.745705, 0.675463, 0.598472, 0.515501, 0.42738, 0.334988,
 0.239249, 0.14112, 0.0415807, -0.0583741, -0.157746, -0.255541, -0.350783,
 -0.44252, -0.529836, -0.611858, -0.687766, -0.756802, -0.818277, -0.871576,
 -0.916166, -0.951602, -0.97753, -0.993691, -0.999923, -0.996165, -0.982453,
 -0.958924, -0.925815, -0.883455, -0.832267, -0.772764, -0.70554, -0.631267,
 -0.550686, -0.464602, -0.373877, -0.279415, -0.182163, -0.0830894}
```

```
In[155]:= ListPlot[Sin[Range[0, 2 π, 0.1]]]
          ListPlot[Sin[Range[0, 2 π, 0.1]], DataRange → {0, 2 π}]
          ListLinePlot[{RandomReal[{-1, 1}, 100], RandomReal[{0.5, 2}, 100]}]
```

Out[155]=



Out[156]=



Out[157]=



Interakció

Input

Az inputba bármilyen *Mathematica* kifejezés beírható.

In[158]:=
```
r = Input["Add meg a kör sugarát", 0];
Print["A kör területe: ", r^2 π, "\nA kör kerülete: ", 2 r π]
Clear[r]
```

```
A kör területe: 25 π
A kör kerülete: 10 π
```

Manipulate

In[161]:=
```
Manipulate[
 Sqrt[a^2 + b^2]
 , {a, 3}, {b, 4}]
```

Out[161]=



In[162]:=
```
Manipulate[Sqrt[a^2 + b^2], {a, Range[5]}, {b, {3, 5, 8}}]
```

Out[162]=



In[163]:=
```
Manipulate[Sqrt[a^2 + b^2], {a, 0, 5, 0.1}, {b, 0, 4, 0.5}]
```

Out[163]=



In[164]:=
```
Manipulate[
 Graphics[
  {color, EdgeForm[If[showEdge, {Thick, Black}, {}]], Disk[p]},
      PlotRange → 5
 ],
 {showEdge, {True, False}}, {color, Blue}, {p, {-5, -5}, {5, 5}}]
```

Out[164]=



In[165]:=
```
Manipulate[Graphics[{Line[{{0, 0}, p}], Green, Disk[p]}, PlotRange → 5],
 {{p, {1, 1}}, Locator}]
```

Out[165]=



In[166]:=
```
Manipulate[Plot[m x + b, {x, -2, 2}, PlotRange → 2],
 {{m, 1, "Meredekség"}, -2, 2}, {{b, 0, "Tengelymetszet"}, -2, 2}]
```

Out[166]=



In[167]:=
```
Manipulate[Plot[f[x], {x, 0, 2 Pi}], {f, {Sin, Cos, Tan, Cot}}]
```

Out[167]=



In[168]:=
```
Manipulate[Expand[(x + 1 / 2)^n], {n, 0, 10, 1}]
```

Out[168]=